

Verweisfunktionen

Excel 2019

Dagmar Serb

V.01

1	VERWEIS-Funktionen	1
1.1	SVERWEIS.....	1
1.1.1	Die 4 ARGUMENTE von SVERWEIS.....	1
1.1.2	Übung: SVerweis	3
1.1.3	Übung: SVerweis mit absolutem Bezug	4
1.1.4	Mögliche Fehlerquellen bei SVerweis	5
1.2	WVERWEIS.....	5
1.3	VERGLEICH	5
1.4	WVERWEIS + VERGLEICH	6
1.5	SVERWEIS + VERGLEICH.....	7

1 VERWEIS-Funktionen

Verweisfunktionen durchsuchen Bereiche nach Suchkriterien. Hantiert man mit großen Datenmengen kann man also mithilfe von Verweisen verschiedene **Datentabellen miteinander in Beziehung setzen**.

1.1 SVERWEIS

SVerweis ist eine Funktion, die Sie **zum Auffinden von Einträgen in einer Datenliste** verwenden können. Das „S“ im Namen SVerweis steht für „senkrecht“ – das bedeutet, dass für das Ausführen dieser Funktion die Daten **spaltenweise angeordnet sein müssen**.

Funktionssyntax

=SVERWEIS(Suchkriterium;Matrix;Spaltenindex;[Bereich_Verweis])

In diesem Beispiel wird durch Eingabe einer Artikelnummer in Zelle G2 der zugehörige Einzelpreis in Zelle G3 ausgegeben:

	A	B	C	D	E	F	G
1							
2	Art.-Nr.	Ersatzteil	Einzelpreis	Status		Art.-Nr.	A003
3	A001	Kühlwasserpumpe	68,39 €	auf Lager		Einzelpreis	15,49 €
4	A002	Lichtmaschine	380,73 €	auf Lager			
5	A003	Luftfilter	15,49 €	auf Lager			
6	A004	Radlager	35,16 €	auf Lager			
7	A005	Schalldämpfer	160,23 €	auf Lager			

Wie arbeitet diese Funktion:

- ▶ In **Zelle G2** wird das **Suchkriterium**, die **Artikelnummer**, **einggegeben**.
- ▶ Excel **durchsucht** nun die **erste Spalte** der **Nachschlagetabelle (Matrix)**, A3:D52, und findet eine **Übereinstimmung**.
- ▶ Der nächste Schritt ist die **spaltenweise (von links nach rechts) Suche** nach dem gewünschten **Suchergebnis** (*Einzelpreis in Spalte 3*), welches schließlich **in Zelle G3** **ausgegeben** wird.

1.1.1 Die 4 ARGUMENTE von SVERWEIS

1. SUCHKRITERIUM (Nachschlagewert)

Welcher Wert soll gesucht werden? In unserem Beispiel suchen wir nach der Artikel-Nr. in Zelle G2. Excel versucht, **in der ersten Spalte der Nachschlagetabelle** eine Übereinstimmung für diesen Wert zu finden. (*Physikalisch ist es irrelevant, wo die erste Spalte der Nachschlagetabelle beginnt, es kann beispielsweise auch die Spalte K sein!*).

=SVERWEIS(G2; Matrix; Spaltenindex; [Bereich_Verweis])

Suchkriterium



2. MATRIX (Nachschlagetabelle)

Matrix bezeichnet den **Datenbereich, der durchsucht werden soll**.

=SVERWEIS(G2; **A3:D52**; Spaltenindex; [Bereich_Verweis])

↑
Matrix

3. SPALTENINDEX (Fundort)

Der Spaltenindex ist eine Zahl und gibt an, **in welcher Spalte sich das Suchergebnis befindet**. Zur Ermittlung zählt man alle Spalten beginnend mit der ganz linken bis hin zur Spalte mit dem Suchergebnis. Unsere Tabelle erstreckt sich von Spalte A bis Spalte D, das Suchergebnis „Einzelpreis“ ist in Spalte C enthalten. Wir zählen: Spalte A=1, B=2, C=3. In unserem Beispiel lautet der Spaltenindex also „3“.

=SVERWEIS(G2; A3:D52; **3**; [Bereich_Verweis])

↑
Spaltenindex

4. BEREICH_VERWEIS

Mit diesem Argument legen Sie fest, ob eine **genaue** oder **ungefähre Übereinstimmung** erforderlich ist:

- **FALSCH** = **exakte Übereinstimmung** - geben Sie „FALSCH“ oder „0“ ein.
- **WAHR** = **annähernde Übereinstimmung** - geben Sie „WAHR“ oder „1“ ein.

In unserem Beispiel benötigen wir eine exakte Übereinstimmung, also Wert „falsch“:

=SVERWEIS(G2; A3:D52; 3; **FALSCH**)

↑
Bereich_Verweis

⚠ Beachten Sie:

Führen Sie kein Argument an, übernimmt Excel **standardmäßig den Wert „WAHR“!**

Beim Wert „WAHR“ muss **die erste Spalte ihrer Tabelle in aufsteigender Reihenfolge sortiert** sein, sonst kommt es zu Fehlern!

WAS ... suchen Sie? - **Suchkriterium**

WO ... soll gesucht werden? - **Matrix (Nachschlagetabelle)**

WELCHE ... Spalte enthält das Suchergebnis? - **Spaltenindex**

Merkhilfe

1.1.2 Übung: SVerweis

- Öffnen Sie die Übungsdatei Funktionen-Verweise.xlsx und gehen Sie zum Tabellenblatt Ersatzteile. Geben Sie die soeben erläuterte Funktion nun selbst ein:
- Positionieren Sie den **Cursor in Zelle G3**.
- Leiten sie die Funktion SVerweis mit „=“ ein.
- Tippen Sie „sv“ ein. **AutoVervollständigung** schlägt Ihnen die **Funktion SVERWEIS** vor.

Art.-Nr.	
Einzelpreis	=sv

- Übernehmen Sie durch **Doppelklick auf SVERWEIS** diese Funktion.

Art.-Nr.	
Einzelpreis	=SVERWEIS(

SVERWEIS(Suchkriterium; Matrix; Spaltenindex; [Bereich_Verweis])

- Geben Sie der Reihe nach die nötigen **Argumente** ein:
 - Unser **Suchkriterium** ist die **Art.-Nr.**, welche wir bei der Abfrage in die **Spalte G2** eingeben werden. **Klicken** Sie Daher auf die **Zelle G2**.

Art.-Nr.	
Einzelpreis	=SVERWEIS(G2

SVERWEIS(Suchkriterium; Matrix; Spaltenindex; [Bereich_Verweis])

- Vergessen Sie den **trennenden Strichpunkt** nicht, bevor Sie nun die **Matrix** eingeben. Der **Zellbereich unserer Nachschlagetabelle** erstreckt sich **von Zelle A3 bis Zelle D52** (ohne Spaltenüberschrift!). Daher A3:D52

Art.-Nr.	
Einzelpreis	=SVERWEIS(G2;A3:D52;

SVERWEIS(Suchkriterium; Matrix; Spaltenindex; [Bereich_Verweis])

- Das nächste Argument ist der **Spaltenindex**. Wir **zählen von Spalte A bis zu Spalte C**, wo sich unser **Suchergebnis, der Einzelpreis**, befindet - also 3

	1 A	2 B	3 C	D	E	F	G	H	I
1									
2	Art.-Nr.	Ersatzteil	Einzelpreis	Status		Art.-Nr.			
3	A001	Kühlwasserpumpe	68,39 €	auf Lager		Einzelpreis	=SVERWEIS(G2;A3:D52;3;		
4	A002	Lichtmaschine	380,73 €	auf Lager			SVERWEIS(Suchkriterium; Matrix; Spaltenindex;		

- Mit dem letzten Argument (**Bereich_Verweis**) geben wir an, ob eine genaue oder ungefähre Übereinstimmung nötig ist. Wir benötigen eine **exakte Übereinstimmung**, daher: falsch

Art.-Nr.	
Einzelpreis	=SVERWEIS(G2;A3:D52;3;falsch

SVERWEIS(Suchkriterium; Matrix; Spaltenindex; [Bereich_Verweis])

FALSCH - Genaue Übereinstimmung

- Beenden Sie die Formel mit „)“.
- Drücken Sie zur Bestätigung **ENTER**.

7. **Probieren Sie Ihre Funktion aus**, indem Sie in **Zelle G2** eine **Artikelnummer** eingeben.

1.1.3 Übung: SVerweis mit absolutem Bezug

1. Wechseln Sie in der aktuellen Übungsdatei auf das Tabellenblatt **Steuertabelle**. Sie haben eine Tabelle mit 9 Mitarbeitern und deren Gehältern. Da die Gehälter unterschiedlich hoch sind, differieren auch deren Steuersätze. Diese gilt es zu ermitteln.


	A	B	C	D	E	F	G
1	Name	Gehalt	Steuersatz			Steuertabelle	
2	Bauer	43.999,00 €	8%				
3	Kiefer	13.242,00 €	2%				
4	Bottler	98.732,00 €	19%			Gehaltsstufe	Prozentsatz
5	Peters	45.673,00 €	9%			- €	0%
6	Petzheim	5.429,00 €	1%			5.000,00 €	1%
7	Hartmann	8.232,00 €	1%			10.000,00 €	2%
8	Breuer	78.109,00 €	15%			15.000,00 €	3%
9	Hassler	39.267,00 €	7%			20.000,00 €	4%
10	Phillips	28.909,00 €	5%			25.000,00 €	5%
11						30.000,00 €	6%
12						35.000,00 €	7%
13						40.000,00 €	8%
14						45.000,00 €	
15							
16							



Beachten Sie:

Die Funktion wird nach unten kopiert– die **Nachschlagetabelle** muss daher als **absoluter Bezug (Dollarzeichen)** angeschrieben werden. Neben dieser Variante gibt es noch eine **elegantere Methode, einen Bezug absolut anzuschreiben: Durch Zuweisen eines Namens!**


2. **Markieren** Sie den Zellbereich **F5:G24**.
3. Setzen Sie den Cursor in das **Namenfeld** und geben Sie dort „Steuer“ ein.



Beginnen Sie jetzt mit der Funktionseingabe:

4. Klicken Sie in **Zelle C2**.
5. Geben Sie „=sv“ ein.
6. AutoVervollständigung schlägt Ihnen die **Funktion SVERWEIS** vor, die Sie **durch Doppelklick übernehmen**.
7. Es folgt die Eingabe des 1. Arguments, dem **Suchkriterium** „Gehalt“: B2
8. Zweites Argument ist die **Matrix**, der wir zuvor einen Namen zugewiesen haben: **Steuer**
9. Nächstes Argument ist der **Spaltenindex** für unser Suchergebnis „Prozentsatz“. Dieser befindet sich in der zweiten Spalte der Matrix, also 2.
10. Im letzten Argument legen wir mit der Eingabe **Wahr** die **ungefähre Übereinstimmung** fest. Dies ist nötig, da die Gehälter ja Gehaltsstufen und keinen exakten Werten zugeordnet werden.

Die fertige Formel:



1.1.4 Mögliche Fehlerquellen bei SVerweis

- › Der **Nachschlagewert** befindet sich **nicht in der ersten Spalte** der Tabelle.
- › **Unnötige Leerzeichen, geschützte Leerzeichen** oder **andere Sonderzeichen** befinden sich in der ersten Spalte der Matrix.
- › Der Wert „**Wahr**“ wird verwendet, ohne die **erste Spalte aufsteigend sortiert** zu haben.
- › Beim **Kopieren der Funktion nach unten** wurde vergessen, dass die Matrix einen **absoluten Bezug** haben muss.

1.2 WVERWEIS

WVerweis ist eine Funktion, die Sie **zum Auffinden von Einträgen in einer Datenliste** verwenden können. Das „**W**“ im Namen WVerweis steht für „**Waagrecht**“ – das bedeutet, dass für das Ausführen dieser Funktion die Daten **zeilenweise angeordnet sein müssen**.

Funktionsyntax	
=	WVERWEIS(Suchkriterium;Matrix;Zeilenindex;[Bereich_Verweis])

WVERWEIS ist der **waagrechte Bruder von SVERWEIS** und agiert analog zu SVERWEIS!

Im Tabellenblatt Auswertung wird der Wert aus Zeile 3 von „Mona“ ausgegeben:

	A	B	C	D	E
1					
2		Lisa	Mona	Rudi	
3		24	3	25	
4		6	14	14	
5		21	47	6	
6					
7		Mona	14		

- › Das **Suchkriterium** bezieht sich auf die Eingabe in Zelle B7.
- › Die **Matrix** erstreckt sich von B2 bis D5.
- › **Zeilenindex** ist 3, da der Wert aus Zeile 3 ausgegeben werden soll.
- › **Bereich_Verweis** ist falsch, da nach einem genauen Ergebnis gesucht wird.

1.3 VERGLEICH

Vergleich wird häufig als **Hilfsfunktion für andere Funktionen** benötigt. Diese Funktion **sucht in einer Matrix den Suchwert** und gibt die **Position des Wertes** zurück.



Funktionssyntax

=VERGLEICH(Suchkriterium;Suchmatrix;[Vergleichstyp])

Argumente	Beschreibung
Suchkriterium (erforderlich)	Wert, mit dem Elemente in der Suchmatrix abgeglichen werden. Suchkriterium kann eine Zahl, Text, logischer Wert oder ein Zellbezug sein.
Suchmatrix (erforderlich)	Bereich, der durchsucht wird.
Vergleichstyp (erforderlich)	Gibt an, auf welche Weise die Werte in der Suchmatrix mit dem Wert für Suchkriterium abglichen.

Vergleichstyp	Verhalten
1 oder nicht angegeben	VERGLEICH sucht nach dem größten Wert, der kleiner oder gleich dem Wert für Suchkriterium ist. Die Werte im Argument Suchmatrix müssen in aufsteigender Reihenfolge angeordnet sein.
0	VERGLEICH sucht nach dem ersten Wert, der mit dem Wert für Suchkriterium genau übereinstimmt. Die Werte im Argument Suchmatrix dürfen in beliebiger Reihenfolge angeordnet sein.
-1	VERGLEICH sucht nach dem kleinsten Wert, der größer oder gleich dem Wert für Suchkriterium ist. Die Werte im Argument Suchmatrix müssen in absteigender Reihenfolge angeordnet sein.

Beispiele Tabellenblatt VERGLEICH Lösung:

	A	B	C	D	E	F
1	Produkt	Anzahl		Vergleich	Formel	Beschreibung
2	Bleistift	26		2	=VERGLEICH(37;B2:B5;1)	Gesucht wird die Position des nächstkleineren Werts von 37 im Bereich B2:B5. Das ist die Zahl 35 an 2. Stelle, daher Ergebnis "2".
3	Füllfeder	35		3	=VERGLEICH(40;B2:B5;0)	Gesucht wird die Position von 40 im Bereich B2:B5. Der Wert befindet sich an 3. Stelle, daher Ergebnis "3".
4	Radiergummi	40	#NV		=VERGLEICH(41;B2:B5;-1)	Gibt einen Fehlerwert zurück, da die Werte im Bereich B2:B5 in absteigender Reihenfolge sortiert sind!
5	Lineal	42				

1.4 WVERWEIS + VERGLEICH

Beispiel: Sie möchten aus nachfolgender Tabelle den Preis inkl. Versand nach US für Artikelnummer 50-XY-03 ermitteln:

	A	B	C	D	E	F	G	H
1	Art.-Nr.	EK	Versand Ö	Versand EU	Versand US		Preis:	379,5
2	50-XY-01	€ 150,00	€ 157,50	€ 154,50	€ 165,00			
3	50-XY-02	€ 224,00	€ 235,20	€ 230,72	€ 246,40			
4	50-XY-03	€ 345,00	€ 362,25	€ 355,35	€ 379,50			
5	50-XY-04	€ 367,00	€ 385,35	€ 378,01	€ 403,70			
6	50-XY-05	€ 691,00	€ 725,55	€ 711,73	€ 760,10			
7	50-XY-06	€ 749,00	€ 786,45	€ 771,47	€ 823,90			

Mit WVERWEIS lösen Sie die Aufgabe wie folgt: =WVERWEIS(E1;C1:E7;4;Falsch)

- Suchkriterium = E1 (Versand US)
- Matrix = C1:E7
- Zeilenindex = 4, da sich die gesuchte Art.-Nr. in Zeile 4 befindet.
- Bereich_Verweis = 0, da eine genaue Übereinstimmung benötigt wird.

Es ist jedoch mühsam und zeitaufwendig, die Zeile, in der sich das Suchergebnis befindet (Zeilenindex), selbst aus der Matrix zu ermitteln. Vor allem bei umfangreichen Matrizen!

Lösung: Den Zeilenindex mit der Funktion VERGLEICH ermitteln (Tabellenblatt WVERWEIS+VERGLEICH Lösung).

H3 =WVERWEIS(H2;C1:E7;VERGLEICH(H1;A1:A7;0);FALSCH)

	A	B	C	D	E	F	G	H
1	Art.-Nr.	EK	Versand Ö	Versand EU	Versand US		Art.-Nr.:	50-XY-03
2	50-XY-01	€ 150,00	€ 157,50	€ 154,50	€ 165,00		Best.-Ort:	Versand US
3	50-XY-02	€ 224,00	€ 235,20	€ 230,72	€ 246,40		Preis:	379,5
4	50-XY-03	€ 345,00	€ 362,25	€ 355,35	€ 379,50			
5	50-XY-04	€ 367,00	€ 385,35	€ 378,01	€ 403,70			
6	50-XY-05	€ 691,00	€ 725,55	€ 711,73	€ 760,10			
7	50-XY-06	€ 749,00	€ 786,45	€ 771,47	€ 823,90			

Tipp:
Am besten, Sie legen sich eine kleine Ergebnistabelle zurecht, bei der Sie die Daten flexibel abfragen können!

Mit WVERWEIS + VERGLEICH lösen Sie die Aufgabe wie folgt: =WVERWEIS(H2;C1:E7;VERGLEICH(H1;A1:A7;0);0)

- Suchkriterium = H2 (Versand US)
- Matrix = C1:E7
- Zeilenindex wird mit VERGLEICH ermittelt:
 - Suchkriterium = H1
 - Matrix = A1:A7
 - Vergleichstyp = 0
- Bereich_Verweis = falsch, da eine genaue Übereinstimmung benötigt wird.

1.5 SVERWEIS + VERGLEICH

Nochmals als Beispiel die Lagerliste - durch Eingabe einer Artikelnummer werden die zugehörigen Informationen ausgegeben (Tabellenblatt SVERWEIS + VERGLEICH-a):

G3 =SVERWEIS(\$G\$2;\$A\$3:\$D\$52;2;FALSCH)

	A	B	C	D	E	F	G
1							SVERWEIS
2	Art.-Nr.	Ersatzteil	Einzelpreis	Status		Art.-Nr.	A002
3	A001	Kühlwasserpumpe	68,39 €	auf Lager		Ersatzteil	Lichtmaschine
4	A002	Lichtmaschine	380,73 €	auf Lager		Einzelpreis	€ 380,73
5	A003	Luftfilter	15,49 €	auf Lager		Status	auf Lager
6	A004	Radlager	35,16 €	auf Lager			
7	A005	Schalldämpfer	160,23 €	auf Lager			
8	A006	Ölwanne	101,89 €	nicht auf Lager			
9	A007	Bremsbeläge	65,99 €	auf Lager			



Mit SVERWEIS lösen Sie die Aufgabe wie folgt:
 =SVERWEIS(\$G\$2;\$A\$3:\$D\$52;2;FALSCH)

- Suchkriterium = G2 (Zelle, wo die Art.-Nr. eingegeben wird)
- Matrix = A3:D52
- Zeilenindex = 2, da sich das gesuchte Ersatzteil in Spalte 2 befindet.
- Bereich_Verweis = falsch, da eine genaue Übereinstimmung benötigt wird.

👁 **Hinweis:** Die Formel verwendet absolute Bezüge („\$G\$2“), weil sie für die weiteren Abfragen nach unten kopiert wird!

⚠ **Beachten Sie:**
 Wird nachträglich innerhalb der Liste eine Spalte eingefügt oder ändern sich die Positionen, wird die Auswertung fehlerhaft:

	A	B	C	D	E	F	G	H
1								SVERWEIS
2	Art.-Nr.	Ersatzteil		Einzelpreis	Status		Art.-Nr.	A002
3	A001	Kühlwasserpumpe		68,39 €	auf Lager		Ersatzteil	✓ Lichtmaschine
4	A002	Lichtmaschine		380,73 €	auf Lager		Einzelpreis	€ ✗ -
5	A003	Luftfilter		15,49 €	auf Lager		Status	€ ✗ 380,73
6	A004	Radlager		35,16 €	auf Lager			
7	A005	Schalldämpfer		160,23 €	auf Lager			

- › Der **Spaltenindex** von „Ersatzteil“ ist mit „2“ definiert.
 =SVERWEIS(\$G\$2;\$A\$3:\$D\$52;**2**;FALSCH)
 D.h. der **Wert aus Spalte 2** der Lagerliste - „Lichtmaschine“ - wird ausgegeben, das ist **korrekt**.
- › Der **Spaltenindex** von „Einzelpreis“ ist mit „3“ definiert.
 =SVERWEIS(\$G\$2;\$A\$3:\$D\$52;**3**;FALSCH)
 D.h. der **Wert aus Spalte 3** der Lagerliste wird ausgegeben. Nachdem sich **an dieser Stelle jetzt die neu eingefügte Spalte befindet**, wird „-“ ausgegeben, das ist **falsch!**
- › Der **Spaltenindex** von „Status“ ist mit „4“ definiert.
 =SVERWEIS(\$G\$2;\$A\$3:\$D\$52;**4**;FALSCH)
 D.h. der **Wert aus Spalte 4** der Lagerliste wird ausgegeben. Nachdem sich **nach Einfügen der neuen Spalte „Status“ auf die 5. Stelle verschoben** hat, wird „380,73 €“ ausgegeben, das ist **falsch!**

💡 **Lösung:** Sie können die **Spaltenindexe nachjustieren**,

oder

die **Spaltenindexe** von vornherein über die **Funktion VERGLEICH** definieren, dann bleibt die Formel konsistent:

	A	B	C	D	E	F	G	J
1								SVERWEIS + VERGLEICH
2	Art.-Nr.	Ersatzteil		Einzelpreis	Status		Art.-Nr.	
3	A001	Kühlwasserpumpe		68,39 €	auf Lager		Ersatzteil	✓ Lichtmaschine
4	A002	Lichtmaschine		380,73 €	auf Lager		Einzelpreis	✓ € 380,73
5	A003	Luftfilter		15,49 €	auf Lager		Status	✓ auf Lager
6	A004	Radlager		35,16 €	auf Lager			
7	A005	Schalldämpfer		160,23 €	auf Lager			

Mit **SVERWEIS** + **VERGLEICH** lösen Sie die Aufgabe wie folgt:

=SVERWEIS(\$G\$2;\$A\$3:\$D\$52;VERGLEICH(F3;\$A\$2:\$D\$2;0);FALSCH)

- Suchkriterium = G2 (Zelle, wo die Art.-Nr. eingegeben wird)
- Matrix = A3:E52
- Zeilenindex wird mit **VERGLEICH** ermittelt:
 - Suchkriterium = F3
 - Matrix = A2:D2
 - Vergleichstyp = 0
- Bereich_Verweis = falsch, da eine genaue Übereinstimmung benötigt wird.

Die eleganteste Lösung allerdings ist,

- › die Liste **als Tabelle zu formatieren**,
[s. Schulungsunterlage MS Excel 2019/Grundlagen/Tabelle definieren],
- › dieser einen **Namen** zu vergeben,
(REGISTER „ENTWURF“ → GRUPPE „EIGENSCHAFTEN“ → „TABELLENNAME“)
- › und dann die vorhin beschriebene Formel mit **SVERWEIS** + **VERGLEICH** anzuwenden.

Damit werden in der Formel auch neu hinzugefügte, der Tabelle angrenzende, Zeilen und Spalten automatisch berücksichtigt!

The screenshot shows the Excel interface with the 'Entwurf' (Design) tab active. The 'Lagerliste' table is defined, and its name is shown in the 'Tabellenname' field. The formula bar displays the formula: `=SVERWEIS(H2;Lagerliste;VERGLEICH(G6;Lagerliste[#Kopfzeilen];0))`. The table structure is as follows:

Art.-Nr.	Ersatzteil	Einzelpreis	Status	neue Spalte	Art.-Nr.	A015
A001	Kühlwasserpumpe	68,39 €	auf Lager		Ersatzteil	ÖlfILTER
A002	Lichtmaschine	380,73 €	auf Lager		Einzelpreis	4,39 €
A003	Luftfilter	15,49 €	auf Lager		Status	auf Lager
A004	Radlager	35,16 €	auf Lager		neue Spalte	test
A005	Schalldämpfer	160,23 €	auf Lager			
A006	Ölwanne	101,89 €	nicht auf Lager			
A007	Bremsbeläge	65,99 €	auf Lager			
A008	Bremscheiben	85,73 €	nicht auf Lager			
A009	Scheinwerfer	35,19 €	auf Lager			
A010	Bremsseil	15,49 €	auf Lager			
A011	Lenkstange	45,29 €	auf Lager			
A012	Antriebswelle	185,36 €	auf Lager			
A013	Gelenkmanschettensatz	26,27 €	auf Lager			
A014	Ölpumpe	46,99 €	auf Lager			
A015	ÖlfILTER	4,39 €	auf Lager	test		
A016	Kraftstofffilter	12,89 €	auf Lager			
A017	Sourstangenkoopf	18,73 €	auf Lager			

Tabellenblatt SVERWEIS + VERGLEICH b

